日本数学教育学会高専部会研究論文誌 vol.5 No.1 *May 1998*

和算の三角形問題のMNR法による数式処理的解法

高遠 節夫*

Solving Triangle Problems in Japanese Mathematics by the Use of Symbolic Computation Systems: The MNR Method

Setsuo TAKATO *

Abstract : Many mathematics educators and researchers have tried to use symbolic computation systems in various ways in mathematical education, and this paper, too, shows how we solve some problems in Japanese geometry, especially those which appear in "sangaku", that is, wooden tablets which were hung in Japanese temples and shrines and on which mathematical problems and figures were drawn. We use the methods of analytic geometry and trigonometry, but triangle problems are difficult to solve in a simple way because the simultaneous equations are non-linear and rapidly become complicated as the number of equations increases. We have developed a method in which we represent various quantities of a triangle (the edges, coordinates of the vertices and so on) using the two tangents of the base angles and the radius of the inscribed circle as a basis. Because the quantities have rational expressions, this method, which we call the MNR method, is effective in solving many triangle problems by the use of symbolic computation systems. Several solutions are given as examples.
Key Words : Symbolic Computation System, Japanese Mathematics, Elementary Geometry

1. はじめに

数式処理システム（以下、数式処理）を教育の場でどのように活用するかについて、これまでに多くの実践研究がなされている（高専では[8], [22], [23], [28]）。また、インターネットを通して数式処理を利用する試み（[12], [17], [24]）や問題発見ツールとしての図形ソフトとの併用も模索されている（[18]）。著者も数式処理の教育利用を考えている一人であり、木更津高専で実施されているテーマ別少人数授業「一般特別研究」（3学年必修1単位、以下、特研）において、和算とくに算額の図形問題に対して、問題解決ツールとしての数式処理を活用することを数年来のテーマとしている（[25]）。一松[4]は「今日の数式処理は数値計算、数式計算、図形計算の三位一体化」と論じており、著者の首肯するところであるが、図形問題では、数式計算による証明あるいは計算、適当な数値の代入による検証あるいは描画の準備、および描画（グラフィックス）、のすべてが問題解決のために必要である。図形問題を解く際の前提知識としては、中学校での初等幾何と三角関数程度で十分であり、その点からしても数式処理の題材として適切である。高等学校においては、平成6年度から実施されている教育課程を構成する「数学A」の中に、選択的に学習される内容と

＊ 木更津工業高等専門学校、千葉県
Department of Mathematics, Kisarazu National College of Technology
Kisarazu-shi, 292-0041 Japan

して「平面幾何」が挙げられているが、高専においては平面幾何を通常授業で取り上げることは時間的余裕からして難しい状況である。しかしながら、図形を解析幾何によって記述した上で代数的に扱うことは、専門科目特に機械系、構造系の科目の基礎としても重要であろう。

図形問題を与えられた学生の反応も教育的に興味深いものがある。Morley の定理（図１）はその美しさと証明の難解さで知られているが、情報工学科の実験実習（３学年の専門科目）の授業において、数式処理を選択している学生たちにこの定理の証明を課題としたとき、大部分の学生がそれまでの学習態度とは一変して、パソコンを傍らにしたまま紙上の図に１時間近くも集中したのである。表１にその解法例を示す（数式処理ソフトは Maple V5 R3, R4 学生版、パソコンは Macintosh Performa 588 、以下同じ）。座標幾何と直線の方程式、三角関数の加法定理、連立方程式を用いれば数式処理にとっては容易であるが、それらの知識を総合的に用いて問題解決にあたる必要があり、その点からの教育的意義も大きいと言えるであろう。

表１

```
> #  Morley's theorem
  tn:=proc(ang)
    normal(expand(tan(ang)));
  end:
> ptIS:=proc(m1,P,m2,Q)
    local x,y,eq1,eq2,ans;
    eq1:=y-P[2]=m1*(x-P[1]);
    eq2:=y-Q[2]=m2*(x-Q[1]);
    ans:=solve({eq1,eq2},{x,y});
    normal(subs(ans,[x,y]));
  end:
> len2:=proc(P,Q)
    (Q[1]-P[1])^2+(Q[2]-P[2])^2;
  end:
> B:=[0,0]: C:=[a,0]:
  alpha:=arctan(m): beta:=arctan(n):
> A:=ptIS(tn(3*alpha),B,tn(Pi-3*beta),C):
  P:=ptIS(m,B,-n,C):
  Q:=ptIS(tn(Pi-2*beta),C,tn(2*Pi/3+alpha-2*beta),A):
  R:=ptIS(tn(2*alpha),B,tn(Pi/3+2*alpha-beta),A):
> PQ2:=factor(len2(P,Q));
  QR2:=factor(len2(Q,R));
  RP2:=factor(len2(R,P));
```

$$PQ2 := 12\,\frac{(n^2+1)\,(m^2+1)\,n^2\,a^2\,m^2}{(m+n)^2\,(3\,n\,m-\sqrt{3}\,n-3-\sqrt{3}\,m)^2}$$

$$QR2 := 12\,\frac{(n^2+1)\,(m^2+1)\,n^2\,a^2\,m^2}{(m+n)^2\,(3\,n\,m-\sqrt{3}\,n-3-\sqrt{3}\,m)^2}$$

$$RP2 := 12\,\frac{(n^2+1)\,(m^2+1)\,n^2\,a^2\,m^2}{(m+n)^2\,(3\,n\,m-\sqrt{3}\,n-3-\sqrt{3}\,m)^2}$$



図１

著者の特研では、和算の幾何問題とくに千葉県の算額[2]にある問題を主に取り上げている。その理由は、学生の数学史に対する関心を惹起したいこともあるが、同時に図形の複雑さと計算量の大きさが数式処理の対象として適当であると考えたからである。和算の幾何問題は、補助線１本で解決するというよりむしろ膨大な計算を根気よく実行した後に美しい結果に到達するものが多く、その意味からも、当時の和算家に数式処理の計算力と高専生の学力がどの程度迫ることができるかは

著者の関心の１つである。Morley の定理が結局のところ連立線形方程式であったのに比して、算額の図形問題は接円の関わるものが多く、その計算量は急激に増加する。それでも円だけに関する問題では、外接（内接）の条件式が２次式ではあるが比較的簡単なため、かなりの数の方程式でも解を得ることができる。図２（笠上寺、文化１１年）はその例である。乙円径を与え戊円径を求める問題であるが、対称性等に着目して未知数を減らすことにより、表２のように９個の２次方程式を作ることができ、２分程度の計算時間で解（戊円径 ＝ １／２ 乙円径）が得られる注1)。

表２

```
> # Ryuujouji
  cir:=proc(C1,C2)
    (C2[1]-C1[1])^2+(C2[2]-C1[2])^2-(C2[3]+C1[3])^2
  end:
> ins:=proc(C1,C2)
    (C2[1]-C1[1])^2+(C2[2]-C1[2])^2-(C2[3]-C1[3])^2
  end:
> r2:='r2':
  C0:=[0,-r1+r5,2*r1+r5]:
  C1a:=[0,-2*r1,r1]: C1b:=[0,0,r1]:
  C2:=[x2,-r1,r2]: C2L:=[-x2,-r1,r2]:
  C3:=[x3,y3,r3]: C3L:=[-x3,y3,r3]:
  C4:=[x4,y4,r4]: C4L:=[-x4,y4,r4]:
  C5:=[0,r1+r5,r5]:
> eq1:=cir(C1b,C2):
  eq2:=cir(C1b,C3):
  eq3:=cir(C1b,C4):
  eq4:=cir(C2,C3):
  eq5:=cir(C3,C4):
  eq6:=cir(C4,C5):
  eq7:=ins(C2,C0):
  eq8:=ins(C3,C0):
  eq9:=ins(C4,C0):
> out1:=[solve({seq(eq.i,i=1..9)},
          {x2,x3,y3,x4,y4,r1,r3,r4,r5})]:
> out2:=op(nops(out1),out1):
  assign(out2):
  r5;
```



図２

$$\frac{1}{2}r2$$

これに対し、三角形と円に関する問題では難しさは倍加する。すなわち辺と円の接する条件式が円と円の場合よりも複雑になり、簡単な図形の場合でもそれを解くことが困難になる。著者は、文献[25]で岩井不動の問題注2)を方程式の個数を減らす工夫を施すことにより解が得られることを報告したが、その後、２底角の２等分角の正接、内接円の半径を三角形の決定要素として問題を定式化することを試みた（[26], [27]）。この方法を著者はＭＮＲ法と呼んでいるが、和算に現れる多くの三角形問題がこれによって数式処理の取り扱いやすい形になり、解の得られることが確かめられた。以下、ＭＮＲ法による三角形の諸量の表現およびＭＮＲ法に基づいて作成された Maple パッケージの関数と大域変数について解説し、いくつかの解法例を示したい。

2. ＭＮＲ法による三角形の表現

現在の数式処理は、有理式の計算に比して根号計算と三角関数の計算を苦手とするため、三角形の諸量をできるだけ簡単な有理式で表すことが重要になる。

図３のように内心Ｉを原点にとり

$$\tan\frac{\angle A}{2}=l,\ \tan\frac{\angle B}{2}=m,\ \tan\frac{\angle C}{2}=n$$

$$(mn+nl+lm=1)$$

とおき、内接円の半径を r とする。また角を

$$\angle B=(m)\qquad(=\ 2\tan^{-1}m)$$

のように表すことにする。

ＭＮＲ法は、この m, n , r によって頂点の座標や辺の長さなどの諸量を表そうとする方法であり、実際に計算すると次のようになる。



図３

（１）頂点

$$A\left(\frac{r(n-m)}{1-mn},\frac{r(1+mn)}{1-mn}\right),\quad B\left(-\frac{r}{m},-r\right),\quad C\left(\frac{r}{n},-r\right)$$

（２）辺の長さ

$$a=\frac{r(m+n)}{mn},\quad b=\frac{r(1+n^2)}{n(1-mn)},\quad c=\frac{r(1+m^2)}{m(1-mn)}$$

（３）外心と外接円の半径

$$O_c\left(\frac{r(m-n)}{2mn},\frac{r(m^2+3m^2n^2+n^2-1)}{4mn(1-mn)}\right),\quad R=\frac{r(m^2+1)(n^2+1)}{4mn(1-mn)}$$

（４）重心と垂心

$$G\left(\frac{r(2mn-1)(n-m)}{3mn(1-mn)},\frac{r(3mn-1)}{3(1-mn)}\right),\quad H\left(\frac{r(n-m)}{1-mn},\frac{r(3m^2n^2-m^2-2mn-n^2+1)}{2mn(1-mn)}\right)$$

（５）傍心と傍接円

$$O_a\left(\frac{r(m-n)}{mn},-\frac{r(mn+1)}{mn}\right),\quad O_b\left(\frac{r(1+n^2)}{n(1-mn)},\frac{rm(1+n^2)}{n(1-mn)}\right),\quad O_c\left(\frac{r(1+m^2)}{m(1-mn)},\frac{rn(1+m^2)}{m(1-mn)}\right)$$

$$r_a=\frac{r}{mn},\qquad r_b=\frac{r(m+n)}{n(1-mn)},\qquad r_c=\frac{r(m+n)}{m(1-mn)}$$

（６）面積

$$S=\frac{(m+n)}{mn(1-mn)}r^2,\quad s=\frac{r(m+n)}{mn(1-mn)}$$

これらの諸式がすべて比較的簡単な有理式であり、数式処理の取り扱いやすい形になっていることが重要である。これに対して辺 a, b, c で表した場合

$$m=\sqrt{\frac{(a+b-c)(b+c-a)}{(a+b+c)(c+a-b)}},\quad n=\sqrt{\frac{(b+c-a)(c+a-b)}{(a+b+c)(a+b-c)}},\quad r=\sqrt{\frac{(b+c-a)(c+a-b)(a+b-c)}{4(a+b+c)}}$$

であり、根号を含んだ取り扱いの困難な式になっている。必要に応じて辺のかわりに

$$\sigma=\sqrt{s}=\sqrt{\frac{a+b+c}{2}},\quad \sigma_1=\sqrt{\sigma^2-a},\quad \sigma_2=\sqrt{\sigma^2-b},\quad \sigma_3=\sqrt{\sigma^2-c}$$

を用いることにする注3）。このとき

$$\sigma_1^{\,2}+\sigma_2^{\,2}+\sigma_3^{\,2}=\sigma^2,\quad m=\frac{\sigma_3\sigma_1}{\sigma\sigma_2},\quad n=\frac{\sigma_1\sigma_2}{\sigma s\sigma_3},\quad r=\frac{\sigma_1\sigma_2\sigma_3}{\sigma}$$

$$\sigma=\sqrt{\frac{r(m+n)}{lm(1-mn)}}=\sqrt{\frac{r}{lmn}},\quad \sigma_1=\sqrt{\frac{r}{l}},\quad \sigma_2=\sqrt{\frac{r}{m}},\quad \sigma_3=\sqrt{\frac{r}{n}}$$

$$a=\sigma^2-\sigma_1^{\,2}=\sigma_2^{\,2}+\sigma_3^{\,2},\quad b=\sigma^2-\sigma_2^{\,2}=\sigma_3^{\,2}+\sigma_1^{\,2},\quad c=\sigma^2-\sigma_3^{\,2}=\sigma_1^{\,2}+\sigma_2^{\,2}$$

の関係が成り立つ。

3. 関数ライブラリの作成

ＭＮＲ法に必要な大域変数と関数からなるライブラリを作成する。数式処理ソフトは Maple を用いるが、Mathematica 等でも同様なライブラリの作成が可能であろう。

（１）大域変数

inCenter	内心
vertexTop, vertexLeft, vertexRight	各頂点の座標
edgeBottom, edgeRight, edgeLeft	各辺の長さ
cicumCenter, radiusCircum	外心，外接円の半径
baryCenter, orthoCenter	重心，垂心
exCenter, rad_exCircle	傍心リスト、傍接円リスト

（２）図形に関する関数

putTriangle(m, n, r)	三角形をおく（大域変数に値を代入）
slideTriangle(A, B)	点Ａが点Ｂに重なるように平行移動
rotateTriangle(θ, C)	点Ｃを中心に θ だけ回転
angtn(m)	＝(m)
tnang(θ)	tan($\theta/2$)
ptDirSeg(A, B, t) 注4)	直線 AB 上の点Ｐの座標　(t＝AP / AB)
eqLine(A, B)	直線 AB の方程式（_X, _Y の式)
eqLine(θ, b)	x 軸と θ をなし切片ｂの直線の方程式
intsectLine($line_1$, $line_2$)	直線 $line_1$ と直線 $line_2$ の交点
conCirLine(C, r, $line$)	中心Ｃ、半径 r の円と直線 $line$ が接する条件
cirCirCir(C_1, r_1, C_2, r_2)	２つの円の外接条件
insCirCir(C_1, r_1, C_2, r_2)	円 C_1 が円 C_2 に内接する条件
subsMnrSgm(m, n, r, σ_1, σ_2, σ_3, e)	式ｅの m, n, r を σ_1, σ_2, σ_3 で置き換える
subsMnrEdge(m, n, r, a, b, c, e)	式ｅの m, n, r を a, b, c で置き換える
subsEdgSgm(a, b, c, σ_1, σ_2, σ_3, e)	式ｅの a, b, c を σ_1, σ_2, σ_3 で置き換える
subsSumSgm(σ_1, σ_2, σ_3, σ, e)	式ｅの $\sigma_1^{\,2}+\sigma_2^{\,2}+\sigma_3^{\,2}$ を σ^2 で置き換える
subsOneSgm(σ_1, σ_2, σ_3, σ, e)	式ｅの $\sigma_1^{\,2}$ を $\sigma^2-\sigma_2^{\,2}-\sigma_3^{\,2}$ で置き換える
subsTwoSgm(σ_1, σ_2, σ_3, σ, e)	式ｅの $\sigma_1^{\,2}+\sigma_2^{\,2}$ を $\sigma^2-\sigma_3^{\,2}$ で置き換える
lenSeg(A, B)	線分 AB の長さ
calcvect(exp)	リストをベクトルとみなして計算
plotTriangle(A, B, C, {options})	三角形 ABC を描く
plotLine(A, B, {options})	線分 AB を描く
plotCircle(C, r, {options})	中心Ｃ，半径ｒの円を描く

（3）その他のユーザー関数（Maple 関数を補足する）

find_equation(ans，x)	連立方程式の解のうち x の満たす方程式
subsvalue(x，val，ans)	連立方程式の解の x に値を代入

例として、三角形の底辺を固定して高さを一定に保ったまま頂点を動かしたときの内心の軌跡を求めることにする[注5)]。

表3

```
> # Naisin no Kiseki
  read `:userlib:agtoolV4.m`:
  read `:userlib:nzTriangleV6.m`:
  unassign('a','h'):
> putTriangle(m,n,r):
  ptM:=calcvect((vertexLeft+vertexRight)/2):
  slideTriangle(ptM,[0,0]):
> eq1:=edgeBottom-a:
  eq2:=vertexTop[2]-h:
  eq3:=x-inCenter[1]:
  eq4:=y-inCenter[2]:
> solve({eq1,eq2,eq3,eq4},{m,n,r,y}):
  find_equation(",y):
  ans:=op(nops("),");
```

ライブラリの読み込み

底辺の中点を原点として三角形をおく

底辺を a 、高さを h 、内心を（x， y）とおく

方程式を解き、x， y の関係式を求める

$$ans := a^2 h - 2\,y\,a^2 - 4\,y^2 h - 4\,h\,x^2 + 8\,y\,x^2$$



図4



図5

図4は内心の軌跡を描いたものである。また、図5は同様にして求めた垂心の軌跡であり、この場合は放物線になる。

4. 和算の問題の解法例

いくつかの和算の問題について、ＭＮＲ法を用いた解法を考察する。表示の図は、すべてライブラリのプロット関数によって描かれたものである。

例1）成田山新勝寺の算額より（[2] P83)

図6の2等辺三角形において小円径を与え大円径を求める問題である。ＭＮＲ法では、内心 I を原点にして三角形をおき、頂角に接する円 O_1 および底辺に接する円 O_2, O_3 の外接条件を解けばよい。大円径＝4小円径と同時に $m=1/2$ であることが示される。



図6

表4

```
> #  Naritasan
  read `:userlib:agtoolV5.m`:
  read `:userlib:nzTriangleV6.m`:
> putTriangle(m,m,r):
> pt01:=ptDirSeg(vertexTop,inCenter,r1/r):
  cirCirCir(inCenter,r,pt01,r1);
  eq1:=op(nops("),"):
```

$$4\,(r1\,m^2-r)\,(r1-r\,m^2)$$

```
> pt02:=ptDirSeg(vertexRight,inCenter,r1/r):
  pt03:=[pt02[1]-2*r1,vertexRight[2]+r1]:
> cirCirCir(inCenter,r,pt03,r1);
  eq2:=op(nops("),"):
```

$$(-r+r1)\,(r1+4\,r1\,m+4\,r1\,m^2-r)$$

```
> [solve({eq1,eq2},{r,m})];
  ans:=op(1,");
```

$$\left[\{r=4\,r1,\,m=\frac{1}{2}\},\,\{m=-1,\,r=r1\},\,\{m=\mathrm{RootOf}(2\,\_Z^2+\_Z+1),\,r=r1\,(2\,\mathrm{RootOf}(2\,\_Z^2+\_Z+1)-1)\}\right]$$

$$ans:=\{r=4\,r1,\,m=\frac{1}{2}\}$$

例2）「精要算法」の問題より（[5] P41）

3個の円径を与えて内接円径を求めるものである。そのまま計算すると平方根が出てくるのであらかじめ $R_i=\sqrt{r_i}$ （i =1, 2, 3）とおく。複数個の解から適切なものを選択することが必要となるが、表5では省略した。根号計算と大小（特に正負）の判定は現在の数式処理の苦手とする処理である。



図7

表5

```
> #  Ex 2.3
  read `:userlib:agtoolV5.m`;
  read `:userlib:nzTriangleV6.m`;
> putTriangle(m,n,r):
  pt01:=ptDirSeg(vertexTop,inCenter,R1^2/r):
  pt02:=ptDirSeg(vertexLeft,inCenter,R2^2/r):
  pt03:=ptDirSeg(vertexRight,inCenter,R3^2/r):
> eq1:=factor(numer(cirCirCir(pt01,R1^2,pt02,R2^2))):
  eq2:=factor(numer(cirCirCir(pt02,R2^2,pt03,R3^2))):
  eq3:=factor(numer(cirCirCir(pt03,R3^2,pt01,R1^2))):
> eq1_s:=op(1,eq1):
  eq2_s:=op(2,eq2):
  eq3_s:=op(1,eq3):
> [solve({eq1_s,eq2_s,eq3_s},{m,n,r})]:
  out1:=op(nops("),"):
> find_equation(out1,r):
  op(nops("),"):
  out2:=collect(",r,factor);
```

$$out2:=(-R2\,R3-R3\,R1-R2\,R1)\,r^2+2\,R3\,R2\,R1\,(R2+R1+R3)\,r-2\,R3^2\,R2^2\,R1^2$$

```
> out3:=simplify([solve(out2,r)],symbolic);
```

$$out3:=\left[\frac{(R1+R3+R2+\sqrt{R1^2+R3^2+R2^2})\,R3\,R2\,R1}{R1\,R3+R1\,R2+R2\,R3},\,\frac{(R1+R3+R2-\sqrt{R1^2+R3^2+R2^2})\,R3\,R2\,R1}{R1\,R3+R1\,R2+R2\,R3}\right]$$

例３）白山神社の算額より（[5] P55）

図８で内部の三角形の内接円、３個の傍接円の半径をそれぞれ $r,\ r_1,\ r_2,\ r_3$ とするとき、外側の三角形の内接円の半径が $R=\frac{1}{2}(r_1+r_2+r_3+r_4)$ で求められるというものである。傍接円の共通接線を求めるための計算が複雑になるが、きれいな結果である。辺の長さがすべて有理式であることもわかる。



図８

表６

```
> # Q2.5.4
  read `:userlib:agtoolV5.m`;
  read `:userlib:nzTriangleV6.m`;
> putTriangle(m,n,r):
> pt01:=exCenter[1]: pt02:=exCenter[2]: pt03:=exCenter[3]:
> r1:=rad_exCircle[1]:
  r2:=rad_exCircle[2]:
  r3:=rad_exCircle[3]:
> eq1:=conCirLine(pt02,r2,eqLine(angtn(t),b)):
  eq2:=conCirLine(pt03,r3,eqLine(angtn(t),b)):
  out1:=[solve({eq1,eq2},{t,b})]:
> temp:=op(5,out1):t1:=subs(temp,t): b1:=factor(subs(temp,b)):
> eq1:=conCirLine(pt03,r3,eqLine(angtn(t),b)):
  eq2:=conCirLine(pt01,r1,eqLine(angtn(t),b)):
  out2:=[solve({eq1,eq2},{t,b})]:
> temp:=op(3,out2):
  t2:=subs(temp,t): b2:=factor(subs(temp,b)):
> eq1:=conCirLine(pt01,r1,eqLine(angtn(t),b)):
  eq2:=conCirLine(pt02,r2,eqLine(angtn(t),b)):
  out3:=[solve({eq1,eq2},{t,b})]:
> temp:=op(2,out3):
  t3:=subs(temp,t): b3:=factor(subs(temp,b)):
> line1:=eqLine(angtn(t1),b1):
  line2:=eqLine(angtn(t2),b2):
  line3:=eqLine(angtn(t3),b3):
> P1:=intsectLine(line2,line3):
  P2:=intsectLine(line3,line1):
  P3:=intsectLine(line1,line2):
> factor(simplify(lenSeg(P2,P3),symbolic)):
  P2P3:=-";
```

$$P2P3 := -\frac{r\,(m+n)\,(m^2\,n^2-m^2-2\,m\,n-1-n^2)}{n\,(n-1)\,(1+n)\,(m-1)\,(m+1)\,m}$$

```
> factor(simplify(lenSeg(P3,P1),symbolic)):
  P3P1:=";
```

$$P3P1 := \frac{r\,(n^2+1)^2\,(m^2\,n^2-m^2-2\,m\,n-1-n^2)}{n\,(n-1)\,(1+n)\,(m\,n-1)\,(m+n+m\,n-1)\,(-m-n+m\,n-1)}$$

```
> factor(simplify(lenSeg(P1,P2),symbolic)):
  P1P2:=";
```

$$P1P2 := \frac{r\,(m^2+1)^2\,(m^2\,n^2-m^2-2\,m\,n-1-n^2)}{m\,(m-1)\,(m+1)\,(m\,n-1)\,(m+n+m\,n-1)\,(-m-n+m\,n-1)}$$

```
> R:=simplify(subsMnrEdg(M,N,R,P2P3,P3P1,P1P2,R),symbolic);
```

$$R := \frac{1}{2}\,\frac{r\,(m^2\,n^2-m^2-2\,m\,n-1-n^2)}{n\,(m\,n-1)\,m}$$

```
> simplify(R-(r+r1+r2+r3)/2);
```

$$0$$

5. まとめ

平面幾何を高専の数学カリキュラムにどのように組み入れるかについては今後の検討課題であるが、図形の問題は高専の学生にとって馴染みの深いものであり、高専の教育の中でより効果的に活かす具体的方法について研究することは重要であろう。その場合に、いわゆる「閃き」に頼らずに、対象と記述式との対応を綿密に調べていくことは技術者を目指す高専学生にとって有益な学習過程であろうと著者は考えている。はじめは和算（算額）に関心を示さなかった特研の学生たちが、まず問題を解こうとして解けなかったときに取り組みの姿勢が変わり、次に苦心して自力で正解に至ったときに算額に対し大きな関心を抱くようになる様子を見れば、数学教育において「問題解決」のもつ意味の大きさを感じずにはいられないのである。数学においては「問題発見」も重要であり、たとえば、川上[11] は Cabri を用いた授業実践の中で、内心と外心が「追いかけっこ」をしてそれらが一致したときに三角形が正三角形になることを生徒たちが見つけた、と報告しており、著者もその教育的意義は極めて大きいと考えているが、同時にそれを検証し証明する過程も大切であろう。また、その視点をもって数式処理とパソコンを用いれば、数学教育への有効かつ有意義な活用が可能であると思われる。ＭＮＲ法による図形問題の解法は、当然のことながら限界はあるにしろ、これまで挙げたいくつかの解法例にあるように、多くの複雑な図形問題を扱うことが可能になり、また、和算（算額）という数学史的に興味深い話題を授業に活かすことの支援になると期待される。

謝辞：小山工業高専名誉教授、松崎利雄先生は、和算についてまったくの初学者である著者を温かくご指導して下さり、また花香文庫をはじめ算額について多くの貴重なご助言をいただいた。群馬工業高専の鈴木福蔵氏は、著者の頼みを快く引き受け、群馬の算額についての得難い資料を入手して下さった。お二人に深く感謝いたします。

［注］

1）表１の解法例では、図から乙円の y 座標を求め、方程式の個数を減らしているが、これを未知数とすると現在使用しているシステムでは解が得られなかった。円の問題についてはこの程度が限界と考えられる。円鎖についてはある程度一般的な定理や公式が必要であろう。ＭＮＲ法を用いればより一般な円鎖の問題が扱える。

2）ＭＮＲ法を考えるきっかけとなった問題で、図で AE＝CD と等円の条件から直角三角形を決定するものである。直線と円の接する条件の連立方程式で解を得るのは困難である（[25], [27]）。花香安精 [10] は藤田定資の精要算法の中にある問題を ΔADC に適用して、$\sqrt{5}$ を含む等式を導き、それによって３次方程式を得ている。会田安明も類似題の評を与えており（[13]）、当時の和算家の関心を惹いたことが窺える。



3）通常の s ではなく σ を用いるのは根号計算を避けるためである。

4）三角形の２辺に接する円を表すのに用いられる．内心をＩ，内接円の半径を r とするとき，辺 AB, AC に接する半径 r_a の円の中心は ptDirSeg(A, I, r_a/r) である．

5）大阪教育大で開催された第３０回数学教育論文発表会において飯島康之氏（愛知教育大）から教えられた問題である。

参考文献

[1] 群馬県和算研究会，「群馬の算額」，１９８７

[2] 平山諦監修，「千葉県の算額」，成田山資料館，１９７０

[3] 久松俊一，高遠節夫，「高専の個性化への試み」，高専教育，第１６号・１９９３，ｐｐ１１２～１１９

[4] 一松信，数学とコンピュータ，共立出版，１９９５

[5] 深川英俊，ダン・ペドー，「日本の幾何——何題解けますか？」，森北出版，１９９１

[6] 深川英俊，ダン・ペドー，「日本の数学——何題解けますか？（下）」，森北出版，１９９４

[7] 藤井康生，「算法天生法指南・問題の解説」，大阪教育図書，１９９７

[8] 一山稔之，「数式処理システムによる数学解析」，日数教高専部会研究論文誌，第４巻・第１号・１９９７，ｐｐ．５１～６２

[9] 岩田至康編，「幾何学大辞典１」，槇書店，１９７１

[10] 花香安精，「柴崎八幡社頭額算解術」，千葉県立図書館所蔵

[11] 川上公一，「数学的探求活動を支援する Cabri-Geometry の利用」，日数教第３０回数学教育論文発表会論文集，１９９７，ｐｐ．３７３～３７８

[12] 松本茂樹，「ICME-8/TG4における遠隔数学教育“公開実験”事例」，日数教学会誌，第７９巻・３号，１９９７，ｐｐ．５７～６３

[13] 松崎利雄，「栃木県算額集」，１９６９

[14] 松崎利雄，「茨城の算額」，１９９７

[15] 文部省，「高等学校学習指導要領解説」，１９８９

[16] 文部省，「中学校指導書」，１９８９

[17] 村上温夫，松本茂樹，「インターネット時代の数学教育」，日数教学会誌，第７９巻・第３号，１９９７，ｐｐ．５０～５６

[18] 大西咲文，高橋正，「数学教育における図形ツールと数式処理システムの併用」，京大数理解析研講究録９８６，１９９７，ｐｐ．１５３～１５６

[19] 笹部貞市郎，「幾何学辞典第２版」，聖文社，１９７６

[20] 佐々木正司他，「mathematicaによる音声分析」，高専教育，第１９号・１９９６，ｐｐ．１３８～１４５

[21] Suzuki，F.，A Certain Property of an Isosceles Trapezoid and Its Application to Chain Circle Problems, Mathematics Magazine, Vol. 68, No. 2, 1995, pp. 136–145

[22] 高木和久，「ノートパソコンによる数式処理教育」，日数教高専部会研究論文誌，第２巻・第１号・１９９５，

[23] 高木和久他，「数式処理ソフトを用いた発見的学習」，日数教学会誌，第７８回総会特集号，１９９６，ｐｐ．５９２

[24] 高橋正他，On the Use of Computer Algebra Systems in Mathematics Education，京大数理解析研講究録，９４１，１９９６，ｐｐ．２３９～２４２

[25] 高遠節夫，「問題解決ツールとしての数式処理」，日数教高専部会研究論文誌，第４巻・第１号・１９９７，ｐｐ．５～１４

[26] 高遠節夫，「数式処理による三角形の問題の解法について」，日数教第３０回数学教育論文発表会論文集，１９９７，ｐｐ．２９５～３００

[27] 高遠節夫，「和算に現れる三角形問題の数式処理による解法について」（投稿中）

[28] 臼田昭司他，「特集簡易言語と数式処理ソフトの教育実践と提案」，日本高専学会誌，第２巻・第１号・１９９７，ｐｐ．２１～３４

[29] Wang, D.M., Algebraic Factoring and Geometry Theorem Proving,Lecture Notes in Comput.Sci., 814, Springer, Berlin. 1994, pp.386–400

[30] Wu, W.T., On the Decision Problem and the Mechanization of Theorem-Proving in Elementary Geometry, Contemporary Mathematics, Volume 29, 1984, pp.213–234

[31] Wu, W.T., Mechanical Theorem Proving in Geometries, Springer-Verlag, 1994